
DISCLAIMER 

Certain portions of this document contain information about Progress Software Corporation’s 
plans for future product development and overall business strategies. Such information is 
proprietary and confidential to Progress Software Corporation and may be used by you solely in 
accordance with the terms and conditions specified in the PSDN Online (http://www.psdn.com) 
Terms of Use (http://psdn.progress.com/terms/index.ssp). Progress Software Corporation 
reserves the right, in its sole discretion, to modify or abandon without notice any of the plans 
described herein pertaining to future development and/or business development strategies.  
Any reference to third party software and/or features is intended for illustration purposes only.  
Progress Software Corporation does not endorse or sponsor such third parties or software. 

 

 

This document accompanies one of a series of video presentations on using Visual Designer in OpenEdge 
Architect, and support for GUI for .NET in OpenEdge 10.2. In this session I create an ABL User Control, a 
container for other related controls, make it into a Navigation panel and add it to a form. As always in 
creating a new class, I open the File menu and select New. Then I select ABL User Control. 

I name it NavPanel. The wizard shows you the key difference between a User Control and an Inherited 
Control. An inherited control inherits and extends a single other control class, typically one of the Microsoft 
or Infragistics control classes. So it becomes an instance of that control. By contrast, a User Control 
inherits the built-in class Progress.Windows.UserControl, which is itself a visual container. The idea is 
that you drop  one or more controls onto the User Control container and then they act as a unit. 

 

I describe this one as a navigation panel for browsing data, and click Finish. What you get in design mode 
is an empty container.  

I grab some Microsoft .NET controls to drop onto it. The first is the GroupBox, to act as a visual container 
for the navigation buttons.  

 

I set the Name property to groupBoxNavPanel, which remember is the name of the variable for the 
control instance in the generated code. Then I set the Text property to Navigation to give a visible name to 
the box. And I’ll make an initial attempt at resizing the container and the GroupBox. 

 

Next I get some Buttons for the navigation panel from the Microsoft Controls. The first one is the First 
button. The Text property for a control is normally its visible Label. And to make it look a bit more 
interesting, I’ll reset the Font property to make it larger and bold, and move the button up into the corner.  

I can speed up the rest of the process by copying and pasting this button to create the other three. Here’s 
the completed panel with all four buttons:  

 

Take a look at the code that got generated for me, by right-clicking and selecting View Source or pressing 
F9. Up at the top is the USING statement that tells the compiler where to find the UserControl class. 
Then the INHERITS phrase on the class definition tells the compiler that NavPanel.cls inherits the user 
control container class. And then you see the variable definitions for the GroupBox and all the buttons. 
Giving the controls reasonable names helps to make the generated code more readable. 

USING Progress.Lang.*. 
USING Progress.Windows.UserControl. 
 
CLASS NavPanel INHERITS UserControl:  
  
    DEFINE PRIVATE VARIABLE buttonPrev AS System.Windows.Forms.Button NO-UNDO. 
    DEFINE PRIVATE VARIABLE buttonNext AS System.Windows.Forms.Button NO-UNDO. 
    DEFINE PRIVATE VARIABLE buttonLast AS System.Windows.Forms.Button NO-UNDO. 
    DEFINE PRIVATE VARIABLE buttonFirst AS System.Windows.Forms.Button NO-UNDO. 
    DEFINE PRIVATE VARIABLE components AS System.ComponentModel.IContainer NO-UNDO. 
    DEFINE PRIVATE VARIABLE groupBoxNavPanel AS System.Windows.Forms.GroupBox  
        NO-UNDO. 

 

Looking down further in the code, you can see the NEW statements that create instances of each of the 
controls and place the references into the variables for them.  

METHOD PRIVATE VOID InitializeComponent(  ): 
   
        /* NOTE: The following method is automatically generated. 
           We strongly suggest that the contents of this method only be modified  
           using the Visual Designer to avoid any incompatible modifications. 
           Modifying the contents of this method using a code editor will  
           invalidate any support for this file. */ 
        THIS-OBJECT:groupBoxNavPanel = NEW System.Windows.Forms.GroupBox(). 
        THIS-OBJECT:buttonFirst = NEW System.Windows.Forms.Button(). 
        THIS-OBJECT:buttonPrev = NEW System.Windows.Forms.Button(). 
        THIS-OBJECT:buttonNext = NEW System.Windows.Forms.Button(). 
        THIS-OBJECT:buttonLast = NEW System.Windows.Forms.Button(). 
        THIS-OBJECT:groupBoxNavPanel:SuspendLayout(). 
        THIS-OBJECT:SuspendLayout(). 

 

In the next section of code in InitializeComponent you can see some interesting aspects of the way the 
objects are grouped. Notice that the GroupBox isn’t just a rectangle around the buttons. The buttons are 
actually added to the Controls collection for the GroupBox, which becomes their immediate parent. Notice 
also that properties such as the GroupBox’s Location and Size are actually objects from the 
System.Drawing class. The Point object defines the location of the corner of the control, and the Size 
defines its dimensions.  

/*  */ 
/* groupBoxNavPanel */ 
/*  */ 
THIS-OBJECT:groupBoxNavPanel:Controls:Add(THIS-OBJECT:buttonLast). 
THIS-OBJECT:groupBoxNavPanel:Controls:Add(THIS-OBJECT:buttonNext). 
THIS-OBJECT:groupBoxNavPanel:Controls:Add(THIS-OBJECT:buttonPrev). 
THIS-OBJECT:groupBoxNavPanel:Controls:Add(THIS-OBJECT:buttonFirst). 
THIS-OBJECT:groupBoxNavPanel:Location = NEW System.Drawing.Point(4, 4). 
THIS-OBJECT:groupBoxNavPanel:Name = "groupBoxNavPanel". 
THIS-OBJECT:groupBoxNavPanel:Size = NEW System.Drawing.Size(328, 49). 
THIS-OBJECT:groupBoxNavPanel:TabIndex = 0. 
THIS-OBJECT:groupBoxNavPanel:TabStop = FALSE. 
THIS-OBJECT:groupBoxNavPanel:Text = "Navigation". 

 

The buttons work the same way, plus there’s a Font object assigned to the Font property for each one. 

/*  */ 
/* buttonFirst */ 
THIS-OBJECT:buttonFirst:Font = NEW System.Drawing.Font 
  ("Microsoft Sans Serif", 12, System.Drawing.FontStyle:Bold,  
      System.Drawing.GraphicsUnit:Point, System.Convert:ToByte(0)). 
THIS-OBJECT:buttonFirst:Location = NEW System.Drawing.Point(6, 19). 
THIS-OBJECT:buttonFirst:Name = "buttonFirst". 
THIS-OBJECT:buttonFirst:Size = NEW System.Drawing.Size(75, 23). 
THIS-OBJECT:buttonFirst:TabIndex = 0. 
THIS-OBJECT:buttonFirst:Text = "First". 

After I save and compile what I have so far, I need to add my new User Control to my own controls group. 
I have already created CustomControls as a controls group. (This was shown in the presentation on 
creating a data-bound TreeView as an Inherited Control.) If I right-click on its header, I can select Add 
Controls. 

 

Then I select the ABL Controls tab to see a list of available controls built in ABL, including both Inherited 
Controls and User Controls. I select the NavPanel control, and click OK, and it’s added to my 
CustomControls group so I can use it in the forms that I build. 

Next I need to try out my new control. I open a form with a Customer grid and binding source (built in the 
presentations on Creating a form and a ProBindingSource) so that I can drop the User Control onto it 
to try it out. I resize and rearrange the form to make room for the navigation panel…  

 

I find it now in my CustomControls. I select the NavPanel like any other control, and drag it onto the 
Customer form. 

The list of properties in the Properties View is interesting for what you don’t see. Because the User Control 
is a container, it hides the specifics of everything it contains; you see only the properties of the container. 
This is very different from an Inherited Control, where all of the inherited public properties are visible to the 
control user, who can set any of them when the control is dropped onto a form. By contrast, if you want to 
expose any properties of the controls in a User Control, you have to do so explicitly by defining properties 
in the class.  

To illustrate this, I add a property ButtonLabelText to the NavPanel class in the editor. I make it a public 
property so it shows up on the property list. It’s a true/false value that lets the user determine whether the 
button labels should show the text labels I defined – First, Last, and so forth – or symbols for the 
navigation operations. Remember that a setter for a property takes the value being assigned as an input 
parameter. This one then determines which set of labels to assign to the buttons’ Text, and then sets the 
property itself to the input value. 

DEFINE PUBLIC PROPERTY ButtonLabelText AS LOGICAL NO-UNDO INIT TRUE 
    GET. 
    SET(plLabelText AS LOGICAL): 
        IF plLabelText NE ButtonLabelText THEN 
        DO: 
            IF plLabelText = TRUE THEN 
                 ASSIGN buttonFirst:TEXT = "First" 
                        buttonPrev:TEXT  = "Prev" 
                        buttonNext:TEXT  = "Next" 
                        buttonLast:TEXT  = "Last". 
            ELSE ASSIGN buttonFirst:TEXT = "<<" 
                        buttonPrev:TEXT  = "<" 
                        buttonNext:TEXT  = ">" 
                        buttonLast:TEXT  = ">>". 
            ButtonLabelText = plLabelText. 
        END. 
      END SET. 

 

Back in design mode for the Customer Form, I can look through the NavPanel properties again, and here is 
my new ButtonLabeltext property. I can try out the code in the property's definition by setting it to False. 

When I save the form, it’s re-run in design mode, and I see the symbols on the button labels. I can re-run 
the form, and here’s my Customer form with the alternative button labels.  

 

This is another illustration of how you must define properties for a User Control in order to provide access 
to any of its contents, which gives you complete authority over what users can change when they use the 
container in a larger form.  

I can click on the buttons, but of course nothing happens. I need event handlers for the buttons, and that is 
the subject of the session on navigation panel events.  

In this presentation I showed you how to create a new ABL User Control, drop other controls onto it, add it 
to your own controls group, define properties for it, and use it as part of a larger form. This is all part of 
creating reusable units of user interface and behavior that you can use to provide a consistent look and feel 
throughout your application. 


